

コンピュータ基礎演習

追加課題(3) 数式をプログラムに

理工学部 情報科学科 隅田 康明

sumida@ip.kyusan-u.ac.jp

追加課題について

- 本来の授業内容から削った内容をいくつか出題
 - 14回授業が13回になったことの補填
 - 遠隔授業により削減した内容の提示
- 制作課題まで提出して、余裕があれば取り組みましょう
 - あくまで、やる気のある学生向けの内容
- いくつか出題する追加課題のうち、1つを提出すれば加点
 - 追加課題の提出で最大+5点
 - ただし、合計の評点は100点を上限とする

想定動作端末

- 追加課題は大学のパソコン教室等での動作を想定
 - 所持端末によっては動作しないものもあるので、動作するものを選んで作業すること

追加課題 3

数式をプログラムに

バラ曲線メソッドを作ってみよう

これは単体ではレポートにはならない（他の回との組み合わせ）

円運動（ぐるぐる回る図形）

- 円運動の基本形

```
float angle = 0;
void setup(){
  size(400,400);
}
void draw(){
  float x = cos(radians(angle)) * 100 + 200; //100は半径
  float y = sin(radians(angle)) * 100 + 200; //200は中心座標
  ellipse(x, y, 10, 10);
  angle += 1;
}
```

中心座標を変えると、円運動の中心が変わる


大きくすると円運動の半径が大きくなる
(大きな起動で回るようになる)

角度を変えると円が回る

```
float x = cos(radians(angle)) * 100 + 200;  
float y = sin(radians(angle)) * 100 + 200;
```

半径100, 角度angle の円の座標を求める場合


半径50の円運動の場合



```
float x = cos(radians(angle)) * 50 + 100;  
float y = sin(radians(angle)) * 50 + 50;
```

数式をプログラムに直す


- 規則性のある図形は、数式で作られていることが多い
- 数式をプログラムで表現できるようになると、色々と面白いことが出来るようになる
 - 数学の知識もある程度必要になる
 - 分からなくても何とかなるが、出来ることの幅が広がる



(少なくとも)
デジタルアートで
表現できることが増える！

正葉曲線（バラ曲線）

- $r = a \sin(n\theta)$ で表される曲線
 - a と n を変えることで、バラに似た形のグラフになる



バラ曲線をプログラムに直す

```
float r1 = 0;
float r2 = 0;
void setup(){
  size(400,400);
}
void draw(){
  translate(width/2, height/2);
  float a = sin(radians(r2));
  float x = a * cos(radians(r1)) * 50;
  float y = a * sin(radians(r1)) * 50;
  ellipse(x, y, 1, 1);
  r1 += 1;
  r2 += 2;
}
```

$\text{radians}(r1) \Rightarrow \text{radians}(r1 * 0.3)$ や、 $r1$ や $r2$ に足す数を変えるとパターンが変わる

バラ曲線を拡張 (スーパーバラ曲線)

- 面白い模様が出るかも

```
float r1 = 0; float r2 = 0;
void setup(){
  size(400,400);
}
void draw(){
  translate(width/2, height/2);
  float a = sin(radians(r2)) * 0.8 + 1;
  float x = a * cos(radians(r1)) * 50;
  float y = a * sin(radians(r1)) * 50;
  ellipse(x, y, 1, 1);
  r1 += 1; //赤字の数値を変えるだけでも色々形が変わる
  r2 += 7;
}
```

バラ曲線を動かすには？

- 円形模様と同じように、繰り返し&メソッド化

```
//loopは繰り返し回数,回数が少ないと1周しなくなることに注意
void roseCurve(float cx, float cy, int loop) {
    pushMatrix();
    translate(cx, cy);
    float r1 = 0, r2 = 0;
    for (int i = 0; i < loop; i++) {
        float a = sin(radians(r2)) * 0.8 + 1;
        float x = a * cos(radians(r1)) * 50;
        float y = a * sin(radians(r1)) * 50;
        ellipse(x, y, 1, 1);
        r1 += 1;
        r2 += 7;
    }
    popMatrix();
}
```

曲線図形を並べて表示する

- 数式のパラメータを変えると模様が替わる


```
float x = a * cos(radians(r1*0.6)) * 100;  
float y = a * sin(radians(r1*0.3)) * 100;
```

- r1やr2に足す数によっても変わる
- 横や縦に並べてみると、
パラメータの変化と模様の変化を観察しやすい

4個並べる場合

- パラメータを変えたroseCurve2,3,4を作って呼び出す

```
float r1 = 0;
float r2 = 0;
void setup(){
  size(400,400);
}
void draw(){
  roseCurve(100, 100, 360);
  roseCurve2(300, 100, 360);
  roseCurve3(100, 100, 720);
  roseCurve4(300, 100, 1000);
}
```



strokeWeight(50);
等を入れて線を太くしてみると
少し面白い

参考：サイクロイド

- 円の1点が転がるときの軌跡から作られる曲線
 - 外サイクロイドの式

$$x = (rc + rm) \cos \theta - rm \cos\left(\frac{rc + rm}{rm} \theta\right)$$

$$y = (rc + rm) \sin \theta - rm \sin\left(\frac{rc + rm}{rm} \theta\right)$$

- 内サイクロイドの式

$$x = (rc - rm) \cos \theta + rm \cos\left(\frac{rc - rm}{rm} \theta\right)$$

$$y = (rc - rm) \sin \theta - rm \sin\left(\frac{rc - rm}{rm} \theta\right)$$

外サイクロイドのプログラム

```
float t = 0; //θ
float rc = 0.8; //数値を変えるとパターンが変わる
float rm = 0.1; //数値を変えるとパターンが変わる
void setup(){
  size(400, 400);
}
void draw(){
  translate(width/2, height/2);
  float x = 100 * ((rc + rm) * cos(t) - rm * cos((rc + rm) / rm * t));
  float y = 100 * ((rc + rm) * sin(t) - rm * sin((rc + rm) / rm * t));
  ellipse(x, y, 1, 1);
  t += 0.1;
}
```

内サイクロイドのプログラム

```
float t = 0; //θ
float rc = 0.8; //数値を変えるとパターンが変わる
float rm = 0.1; //数値を変えるとパターンが変わる
void setup(){
  size(400, 400);
}
void draw(){
  translate(width/2, height/2);
  float x = 100 * ((rc - rm) * cos(t) + rm * cos((rc - rm) / rm * t));
  float y = 100 * ((rc - rm) * sin(t) - rm * sin((rc - rm) / rm * t));
  ellipse(x, y, 1, 1);
  t += 0.1;
}
```


追加課題 3 のレポート

- 他の課題と合わせたアレンジした作品を提出
 - バラ曲線を動かしてみる
 - 3Dバラ曲線にしてみる
 - 音で動くバラ曲線
 - バラ曲線を敷き詰める(第9回アレンジ)
- など、作ったバラ曲線を部品として、他で利用しよう